
آشنايي با ساختار و اجزاي برنامه در محيط زبان برنامه​سازي پاسكال – 2

13-1- آرايه​ها
قبل از اين كه به بحث در مورد آرايه​ها بپردازيم، اجازه دهيد ابتدا يك برنامه بنويسيم. برنامه​اي كه سه عدد را در سه متغير ذخيره كرده و معدل آنها را محاسبه كند. نوشتن اين برنامه براي ما كار ساده​اي است و مي​تواند به صورت زير نوشته شود:
 از تابع inputbox براي گرفتن سه عدد ورودي، از يك Label براي چاپ معدل و از يك كليد براي انجام عمليات استفاده مي​كنيم. سپس متن برنامه در رخدادonclick  شيء كليد را به صورت زير مي​نويسيم:
Var
  i,n : integer ;

  sum : integer ;
  avg : real ;

Begin
  sum := 0 ;

  for i := 1 to 3 do

    Begin
      n := strtoint ( inputbox ( 'number enterance' , 'Enter number :' , '' ) );

      sum := sum + n ;

    End ;

  avg := sum / 3 ;

           label1.caption := floattostr ( avg ) ;
        End ;
خوب، اين برنامه معدل سه عدد را براي ما محاسبه مي​كند، ولي اعداد ورودي را جايي نگهداري نمي​كند. در صورتيكه بخوا​هيم اين اعداد را ثبت كنيم، برنامه را بايد به شكل ديگري تغيير دهيم. براي اين كار بايستي تك تك اعداد گرفته شده را در يك متغير قرار دهيم. در اين صورت برنامه به شكل زير تغيير مي​كند:

Var
  a1,a2,a3 : integer ;

  sum      : integer ;

  avg      : real ;

Begin
  a1 := strtoint ( inputbox ( 'number entrance' , 'Enter first  number :' , '' ) );

  a2 := strtoint ( inputbox ( 'number entrance' , 'Enter second number :' , '' ) );

  a3 := strtoint ( inputbox ( 'number entrance' , 'Enter third  number :' , '' ) );

  sum := a1+a2+a3 ;

  avg := sum / 3 ;

  memo1.lines.add ( inttostr ( a1 ) ) ;

  memo1.lines.add ( inttostr ( a2 ) ) ;

  memo1.lines.add ( inttostr ( a3 ) ) ;
  label1.caption := floattostr ( avg ) ;
            End ;
همان​طوري​كه از اين برنامه مشاهده مي​شود، اعداد ورودي در متغيرهايي ذخيره مي​شوند. بنابراين مي​توان آنها را نگهداري، و مجدد از آنها استفاده نمود، در صورتي كه برنامه اول نمي​توانست چنين كاري را براي ما انجام دهد. حال فرض كنيد مي​خواهيم برنامه دوم را بگونه​اي تعميم دهيم كه به جاي سه عدد، معدل سي عدد را محاسبه كند. در اين صورت چه بايد بكنيم؟

اگر بخواهيم با همان روش تعريف يك متغير براي هر عدد، اين كار را انجام دهيم، مسلما خواهيد  گفت كه چنين كاري بسيار دشوار است. چرا كه بايد 30 متغير گرفته و عمليات را بر تك تك آنها انجام دهيم. اينجاست كه نياز به يك متغير پيشرفته​تر داريم كه بتوانيم به راحتي اين كار را انجام دهيم. اين متغير آرايه است.

تعريف آرايه:

آرايه، متغيري است كه به جاي يك خانه، حاوي چند خانه است. به صورتي كه هر يك از خانه​هاي آن با يك شماره يا يك انديس نام​گذاري مي​شود. براي خواندن يا نوشتن يك مقدار در خانه مربوطه، بايد شماره آن خانه را هم ذكر كرد. مثال زير نحوه تعريف يك آرايه را نشان مي​دهد:

Var
   A: array[1..10] of byte ;

      …

در اين مثال يك آرايه با نام A، با تعداد 10 خانه تعريف شده است، كه هر كدام از خانه​ها از نوع بايت هستند. براي دسترسي به هر يك از خانه​هاي آرايه، كافي است شماره آن خانه را در كروشه قرار دهيم، مثلا A[3]، كه به معني خانه شماره 3 از آرايه A است. به دستورات زير توجه نماييد:

     A[1] := 3 ;

     if A[4] = 5 then

دستور اول مقدار 3 را در خانه شماره 1 آرايه قرار مي​دهد و دستور دوم بررسي مي​كند كه آيا خانه شماره 4 آرايه، برابر با مقدار 5 است يا خير.


برخي از خصوصيات مهم آرايه عبارتند از:

· داراي تعدادي (حداقل يك) عنصر است.
· همه عناصر آن از يك نوع هستند.
· هر عنصر از طريق يك شماره قابل دستيابي است.
· شماره عناصر بايد از نوع ordinal باشد. مثلا آرايه با شماره خانه​هاي اعشاري قابل تعريف نيست. اما آرايه با انديس كاراكتر قابل تعريف است، مثلا: 
C : Array['A'..'Z'] of Byte ;

C['D'] := 15 ;
· شماره​ها داراي يك كران پايين و كران بالا هستند.
· كليه شماره​هاي عناصر كران پايين تا بالا ترتيبي هستند. چرا كه محدوده اعداد از عددي شروع شده و بطور پيوسته تا عددي ديگر ادامه پيدا مي​كند. مثلا a[3..10]، آرايه​اي است كه خانه​هاي آن به ترتيب 3و4و...و9و10 هستند.
· تعداد عناصر آرايه بايد از ابتدا مشخص باشند (در هنگام تعريف آرايه).
· نكته قبلي در مورد آرايه​هاي پويا صدق نمي​كند. آرايه​هاي پويا كه يكي از قابليتهاي موجود در زبان دلفي است، آرايه​هايي بدون اندازه و محدوده هستند. در فصول بعدي كتاب، از اين نوع آرايه خواهيم گفت.
13-1-1- حافظه مورد نياز آرايه

مي​دانيم كه هر يك از متغيرها بستگي به نوع​شان، فضايي را در حافظه اشغال مي​كنند. مقدار حافظه​اي كه در اختيار آرايه قرار مي​گيرد ، به طول آرايه و نوع عناصر آن بستگي دارد. به عنوان مثال اگر آرايه​اي از نوع صحيح  به طول 10 داشته باشيم ، 10 × 4  بايت حافظه  به آن  اختصاص مي​يابد، چون نوع صحيح 4 بايت از حافظه اشغال مي​كند.
مثال 13-1

برنامه​اي بنويسيد كه نمرات تعدادي دانشجو را دريافت كرده و معدل آنها را محاسبه كند.
const

  max = 3 ;

Var
  a : array[1..max] of integer ;

  sum , I  : integer ;

  avg      : real ;

Begin
  sum := 0 ;

  for I := 1 to max do

    a[I] := strtoint(Inputbox('Give Numbers','enter number ' + inttostr(i),'')) ;

  for I := 1 to max do

    sum := sum + a[I] ;

  for I := 1 to max do

    memo1.lines.add(inttostr(a[I])) ;

  avg := sum / max ;

  label1.caption := floattostr ( avg ) ;

End ;
همان​طوركه مشاهده مي​كنيد در اين برنامه، تعداد عناصر آرايه را به مقدار ثابت max محدود كرديم. با تغيير مقدار max در متن برنامه مي​توانيم تعداد خانه​هاي آرايه را تغيير دهيم. اما اگر بخواهيم كاري كنيم كه تعداد خانه​هاي آرايه را در حين اجراي برنامه و از ورودي دريافت كنيم، مثلا زماني كه تعداد دانشجويان مشخص نيست و اين تعداد در زمان اجرا، هر عددي ممكن است باشد، در اين صورت با مشكل مواجه مي​شويم. همان طور كه گفتيم تعداد عناصر آرايه بايد از ابتدا مشخص باشند و نمي​توان در زمان اجرا اندازه آرايه را تعيين نمود. در اين مواقع، يا تعداد عناصر آرايه را تا حد ضروري بزرگ مي​گيريم( كه اين باعث اتلاف حافظه مي​شود) و يا از آرايه پويا (ديناميك) استفاده مي​كنيم. آرايه​هاي پويا اندازه​هاي متغيري دارند و مي​توان از آنها در مواقعي كه تعداد اعضاي آرايه از قبل مشخص نيست، استفاده نمود. نحوه استفاده از اين آرايه در فصل 17 ذكر شده است.

تمرين

برنامه فوق را به شكلي توسعه دهيد كه نام دانشجويان كلاس را نيز دريافت كند.
زيربرنامه​اي بنويسيد كه 3 عدد از ورودي دريافت كند و آنها را به ترتيب از كوچك به بزرگ چاپ كند. (سعي كنيد به نحوي تعداد 3 را افزايش دهيد!)

در برنامه زير آرايه​اي از بولين تعريف شده و بر روي آن اعمالي انجام شده است. به نظر شما، با هر بار فشرن كليد 1 و 2، چه مقداري در خروجي چاپ خواهد شد؟

const

  result : array [boolean] of integer = (0,0) ;

Procedure TForm1.BitBtn1Click(SEnder: TObject);

Begin
  result[true] := result[true] + 1 ;

  edit1.text := inttostr ( result[true] ) ;

End;

Procedure TForm1.BitBtn2Click(SEnder: TObject);

Begin
  result[False] := result[False] + 1 ;

  edit2.text := inttostr ( result[False] ) ;

           End;
13-1-2- آرايه دو (چند) بعدي

ديديم كه هر يك از خانه​هاي آرايه مي​تواند هر يك از انواع Ordinal، مانند integer و char و ... باشد. براي دسترسي به هر يك از عناصر اين آرايه​ها، از يك انديس استفاده كرديم، مثلا a[2]. اين آرايه​ها را آرايه​هاي يك​بعدي مي​ناميم. اما آيا مي​توان در هر يك از عناصر آرايه، يك آرايه تعريف كرد! مثلا مي​توان چنين چيزي نوشت؟
A : array [1..10] of array [1..20] of byte ;
البته! اما بدين شكل:

A : array [1..10, 1..20] of byte ;

اين تعريف يك ماتريس 20×10 ايجاد مي​كند كه حاوي ده سطر و بيست ستون است. اين آرايه را يك آرايه دوبعدي مي​ناميم، چرا كه براي دسترسي به عناصر آن بايد از دو انديس استفاده كرد. مثلا a[2,3] يعني عنصري كه در سطر دوم و ستون سوم قرار دارد. آرايه​هاي دوبعدي مانند يك ماتريس هستند. همچنين يك آرايه مي​توان چند بعد نيز داشته باشد. مثلا آرايه تعريف شده زير يك آرايه پنج بعدي است:

Z : array[1..10, 1..20, 5..10, ‘a’..’z’, Boolean] of real ;
توجه كنيد كه انديس ماقبل آخر در اين آرايه از نوع كاراكتري است. يعني خانه​ها آن با حروف ‘a’ تا ‘z’ نام​گذاري شده​اند. دسترسي يك عنصر از اين آرايه مي​تواند به اين صورت باشد:

Const positive = true ; neqative = False ; 

Z[5,3,9,’b’,positive] := 23.2 ;
13-1-3- حافظه آرايه​هاي دو بعدي

آرايه​هاي دو بعدي در دلفي​، در محيط​هاي متوالي حافظه قرار مي​گيرند. آرايه​هاي دو بعدي به صورت سطري ذخيره مي​شوند. كامپايلر براي دستيابي به هر يك از عناصر آرايه، تفاوت مكان (فاصله) آن عنصر را از اولين عنصر محاسبه مي​كند. براي انجام اين محاسبات، كامپايلر بايد اندازه هر عنصر آرايه را بر حسب بايت و تعداد عناصر موجود در هر سطر را بداند. هر يك از اين دو مقدار، از تعريف آرايه مشخص مي​شود. به عنوان مثال، فرض كنيد آرايه​اي به صورت زير تعريف شده باشد:

              Var
             X : array [1..2 , 1..3] of integer ;
اگر محتويات اين آرايه به صورت زير باشد:

	4
	3
	7

	9
	6
	11


در اين صورت اين آرايه در حافظه به شكل زير نمايش داده مي​شود:

مثال 13-2

برنامه​اي بنويسيد كه نمرات دانش​آموزان يك كلاس را ذخيره كند. نمرات هر دانش​آموز شامل نمرات سه ثلث و مجموع و معدل آنهاست و ضمنا هر نمره داراي دو بخش مستقل است. يك بخش مربوط به نمره كلاسي و عملي و يك بخش مربوط به نمره امتحان.

اما اجازه دهيد قبل از نوشتن اين برنامه، چگونگي استفاده از چندين فرم در برنامه را ياد بگيريم. چرا كه در اين مثال، ما از دو فرم استفاده خواهيم كرد؛ يك فرم براي گرفتن مشخصات دانش​آموز جديد و محاسبه معدل، و فرم ديگر براي ورود نمرات دانش​آموز. (شكل 13-1 و شكل 13-2). 

13-2- بكارگيري چندين فرم در برنامه

بسياري از اوقات نياز به استفاده از چندين فرم در برنامه داريم. مثلا در برنامه مثال 13-2 كه قرار است بنويسيم، به دو فرم نياز داريم. يك فرم براي ورود اطلاعات دانشجو، و فرم ديگر براي نمايش اطلاعات. تمامي برنامه​هايي كه تا به حال نوشتيم، همگي داراي يك فرم بودند. براي افزودن يك فرم به برنامه، از منوي File  وارد گزينه New شويد و از آنجا Form را انتخاب كنيد. با اين كار فرم جديدي همراه با يك واحد جديد، به برنامه افزوده مي​شود. به عنوان مثال اگر تا به حال يك فرم و يك واحد داشتيم، پس از New كردن، دو فرم و دو واحد خواهيم داشت. اين كار را مي​توان در صورت نياز براي ايجاد فرمهاي جديد تكرار كرد و به تعداد دلخواه فرم جديد در برنامه ايجاد كرد.
نكته مهمي كه در رابطه با برنامه​هاي شامل چند فرم وجود دارد، چگونگي رفتن از يك فرم به فرمي ديگر، انتخاب يك فرم از بين فرمهايي كه در برنامه ايجاد شده​ است و حذف يك فرم دلخواه از برنامه است. براي ديدن همه فرمها و واحدهاي برنامه، از منوي View، Project Manager را انتخاب كنيد (شكل 13-3). با كمك اين پنجره مي​توان مديريت فرمها و واحدهاي موجود در برنامه را انجام داد. با انتخاب فرم يا واحد مورد نظر از ساختار درختي موجود در اين پنجره، مي​توان آن را مشاهده نمود. 

براي حذف يك واحد از برنامه، وارد منوي Project شده، Remove From Project را انتخاب كنيد. سپس از بين واحدهاي موجود در اين پنجره مي​توانيد واحد مورد نظر را انتخاب كرده و آن را از برنامه حذف كنيد.
حال چگونه در برنامه فرم جديد را بكار گيريم! فرض كنيد كه دو فرم در برنامه ايجاد نموده​ايد و مي​خواهيد از فرم اول، فرم دو را صدا بزنيد. كافي است دستور مربوط به فراخواني فرم را بكار گيريد. اين دستور بدين صورت است:

Form2.show ;
هرگاه اين دستور اجرا شود، فرم شماره 2 فراخواني و نمايش داده مي​شود. به عنوان مثال اگر بخواهيد با فشردن يك كليد فرم شماره 2 نمايان شود، بايد اين دستور را در رخداد OnClick شيء كليد بنويسيد.

اما دستور ديگري نيز براي فراخواني يك فرم در برنامه وجود دارد. اين دستور چنين است:

Form2.showmodal ;
تفاوت اين دستور با دستور قبلي در اين است كه در اين دستور هنگامي كه فرم 2 نمايان شد، تا زمان باز بودن اين فرم، نمي​توان به فرم قبلي بازگشت، مگر اين كه فرم دو بسته شود. اما با دستور اول، مي​توان هنگامي كه فرم دو فراخواني شد، بدون بستن آن، به فرم يك رفت.

ادامه مثال 13-2

حال با توضيحات داده شده، برنامه نمرات دانش​آموزان كلاس را ادامه مي​دهيم. گفتيم كه گرفتن اطلاعات دانش​آموز جديد و محاسبه معدل دانش​آموزان را در واحد1، و اخذ نمرات هر دانش​آموز را در واحد 2 انجام مي​دهيم. بنابراين كافي است تا در واحد يك و در بخش دانش​آموز جديد، فرم 2 را صدا زده و در آنجا نمرات دانش​آموز جديد را دريافت نماييم.
برنامه واحد 1:

……

uses Unit2;

{$R *.DFM}

Var
  sg          : array [1..30,1..5,1..2] of single ;

const

  max_student : byte = 0 ;
Procedure TForm1.Button1Click(Sender: TObject);

Begin
  form2.showmodal ;

  max_student := max_student + 1 ;

  sg[max_student,1,1] := asg[1,1] ;

  sg[max_student,2,1] := asg[2,1] ;

  sg[max_student,3,1] := asg[3,1] ;

  sg[max_student,1,2] := asg[1,2] ;

  sg[max_student,2,2] := asg[2,2] ;

  sg[max_student,3,2] := asg[3,2] ;

End;

Procedure TForm1.Button2Click(SEnder: TObject);

Var
  i,j,k : byte ;

  s     : shortstring ;

Begin
  for i := 1 to max_student do

    Begin // compute a student grades

      for k := 1 to 2 do

        Begin
          sg[i,4,k] := 0 ;

          for j := 1 to 3 do

            sg[i,4,k] := sg[i,4,k] + sg[i,j,k] ;

          sg[i,5,k] := sg[i,4,k] / 3 ;

        End ;

    End ;

  for i := 1 to max_student do

    for k := 1 to 2 do

       Begin
         if k = 1 then

           s := 'Quese '

         else

           s := 'main ' ;

         s := s + 'student['+inttostr ( i )+'] :' ;

         for j := 1 to 5 do

           s := s + floattostr ( sg[i,j,k]) + '  '  ;

         memo1.lines.add ( s ) ;

       End ;

End;

برنامه واحد 2:

Var
  Form2: TForm2;

  asg  : array [1..3,1..2] of single ;

implementation

{$R *.DFM}

Procedure TForm2.BitBtn1Click(SEnder: TObject);

Begin
  asg[1,1] := strtofloat ( edit1.text ) ;

  asg[2,1] := strtofloat ( edit2.text ) ;

  asg[3,1] := strtofloat ( edit3.text ) ;

  asg[1,2] := strtofloat ( edit4.text ) ;

  asg[2,2] := strtofloat ( edit5.text ) ;

  asg[3,2] := strtofloat ( edit6.text ) ;

End;

تمرين

برنامه مثال 13-2​ را به نگهداري اطلاعات دانش​آموزان يك مدرسه تعميم دهيد.
13-3- مرتب كردن يا  Sort
قبل ار بررسي مرتب​كردن و الگوريتمهاي مربوط به آن، ابتدا به مثال 13-3 توجه كنيد. 
مثال 13-3

زيربرنامه​اي​​ بنويسيد كه 3 عدد را از ورودي دريافت كند و آنها را به ترتيب از كوچك به بزرگ چاپ نمايد.

براي اين كار ابتدا از بين سه عدد، عدد كوچكتر را انتخاب مي​كنيم. سپس به دنبال عدد بزرگتر گشته و آن را پيدا مي​كنيم. در انتها با داشتن عدد كوچكتر و عدد بزرگتر، عدد مياني را بدست مي​آوريم.
Procedure TForm1.Button1Click(SEnder: TObject);

const

  max = 3 ;

Var
  a,b   : array [1..max] of integer ;

  h,l,m : integer ;

  i     : integer ;

Begin
  for i :=  1 to max do

    a[i] := strtoint ( inputbox ( '' , '' , '' )) ;

  // finding maximum ( higher ) value of list

  h := a[1] ;

  for i := 2 to max do

    if a[i] > h then

      h := a[i] ;

  // finding minimum ( lower ) value of list

  l := a[1] ;

  for i := 2 to max do

    if a[i] < l then

      l := a[i] ;

  // finding mid value of list

  m := a[1] ;

  for i := 2 to max do

    if ( l < a[i] ) and ( a[i] < h )then

      m := a[i] ;

  // moving results in b

  b[1] := l ;

  b[2] := m ;

  b[3] := h ;

  // printing sorted list

  for i :=  1 to max do

    memo1.lines.add ( inttostr ( b[i] ) ) ;

End;
اگر بخواهيم اين برنامه را تعميم دهيم و مثلا براي 4 عدد اجرا كنيم، الگوريتم برنامه براي چنين كاري مناسب نيست. البته براي پيدا نمودن مقادير ماكزيمم و مينيمم مشكلي وجود ندارد، ولي پيدا نمودن اعداد مياني با اين الگوريتم غيرممكن است. لذا بايستي راه حل بهتري براي اين كار پيدا كنيم. چندين الگوريتم براي اين كار وجود دارد كه به الگوريتمهاي مرتب​سازي يا Sort معروفند. مرتب​سازي در بسياري از برنامه​ها از ضروريات كار برنامه​نويسي است و يك برنامه​نويس معمولا با آن زياد سر و كار دارد.

13-3-1- مرتب​سازي انتخابي يا Selection Sort
در اين روش، نياز به يك آرايه ديگر مشابه آرايه مورد مرتب​سازي داريم. ورودي در آرايه اول و خروجي در آرايه دوم انجام مي​شود. ابتدا بايد بزرگترين عدد آرايه ورودي يافت شود. مقداري به آن اضافه شود (مثلا يكي) كه مطمئن باشيم اين عدد از همه اعداد موجود در آرايه بزرگتر است. سپس بايد طي چند مرحله در آرايه ورودي، بدنبال كوچكترين عدد گشت و آن را بترتيب در آرايه خروجي گذاشت. براي اينكه اين خانه مجددا يافته نشود، آن عدد بزرگ اوليه را (كه از همه اعداد آرايه بزرگتر بود) در خانه كوچك يافت شده در آرايه ورودي قرار مي​دهيم. اين كار بايد به تعداد خانه​هاي آرايه تكرار شود. پس از اتمام كار، آرايه خروجي حاوي اعداد از كوچك به بزرگ است. ولي اطلاعات آرايه ورودي از بين رفته است. شكل 13-4 مثالي از اين مرتب​سازي را نشان مي​دهد.

13-3-2- مرتب​سازي درجي يا Insertion Sort
در اين روش، يك آرايه به عنوان آرايه خروجي در نظر گرفته مي​شود. از ابتداي آرايه ورودي، اعداد تك به تك خوانده مي​شود. هر عنصر در جاي خودش در ميان ليست آرايه خروجي درج مي​شود. اگر هيچ عنصري وجود نداشته باشد، يا تمام عناصر ليست از عدد خوانده شده كوچكتر باشند، عدد در انتهاي ليست درج مي​شود. اما اگر عدد بايد درميان ليست درج شود، به وسيله حركت دادن بقيه اعداد به جلو (اصطلاحا به آن Shift دادن گفته مي​شود) جا براي درج عدد در جاي مناسب خود باز مي​شود و عدد در آنجا درج مي​شود. شكل 13-5 مثالي از اين ليست مرتب​سازي را نشان مي​دهد.

13-3-3- مرتب​سازي حبابي يا Bubble Sort
روش مرتب​سازي حبابي، عمل مرتب​سازي را درجا (در همان آرايه) انجام مي​دهد و نيازي به آرايه اضافي نيست. در مرتب​سازي حبابي، دو عنصر متوالي با هم مقايسه مي​شوند؛ چنانچه اولي از دومي بزرگتر باشد، جاي آنها در آرايه عوض مي​شود. وقتي يك بار تمامي عناصر، دوبدو با هم مقايسه شدند، بزرگترين عنصر به انتهاي آرايه مي​رود. در مرحله بعد كه عناصر آرايه دوبدو با هم مقايسه مي​شوند، دومين عدد از نظر بزرگي، قبل از بزرگترين عنصر، در انتهاي آرايه قرار مي​گيرد. اين روند تا مرتب​سازي نهايي ادامه پيدا مي​كند. شكل 13-6 مثالي از اين مرتب​سازي را نشان مي​دهد.

مثال 13-2

برنامه​اي بنويسيد كه تعدادي عدد را خوانده، آنها را در آرايه​اي قرار دهد و به سه روش گفته شده، آنها را مرتب كند (شكل 13-7).
type

  str_30 = shortstring ;

const

  max_array_len = 5 ;

Var
  names : array [1..max_array_len] of str_30 ;

  a     : array [1..max_array_len] of byte ;
--------------------------------------------------------------

             (* define Array For Sort *)

Procedure TForm1.Button3Click(SEnder: TObject);
Var
  i : integer ;

Begin (* define array by entering its values *)

  for i := 1 to max_array_len do

    Begin
      names[i] :=inputbox ( 'Names Entrance' ,

                            'names['+inttostr ( i )+']:' ,

                            '' ) ;

      a[i] := strtoint ( inputbox ( 'Array Field Entrance' ,
                                    'Field ['+inttostr ( i )+']:' ,

                                    '' )) ;

    End ;

  for i := 1 to max_array_len do

    memo2.lines.add ( names[i]+' :'+inttostr ( a[i] ) ) ;

End;

-----------------------------------------------------------

                        (* selection sort *)

Procedure TForm1.Button2Click(SEnder: TObject);

  Function find_min : byte ;

    Var
      i : integer ;

    Begin
      result := a[1] ;

      for i := 2 to max_array_len do

        if a[i] < result then

          result := a[i] ;

    End ;

  Procedure  find_max ( Var max_val, pos : byte ) ;

    Var
      i : integer ;

    Begin
      max_val := a[1] ;

      pos     := 1 ;

      for i := 2 to max_array_len do

        if a[i] > max_val then

          Begin
            max_val := a[i] ;

            pos     := i ;

          End ;

    End ;

Var
  min,

  max,pos : byte ;

  i       : integer ;

  b       : array [1..max_array_len] of byte ;

Begin  (* selection sort main *)

  min := find_min - 1 ;

  for i := 1 to max_array_len do

    Begin
      find_max ( max , pos ) ;

      b[i] := max ;

      a[pos] := min ;

    End ;

  // copy b to a

  for i := 1 to max_array_len do

    a[i] := b[i] ;

  // printing sorted list

  memo1.lines.add ( '-selection sort result-' ) ;

  for i :=  1 to max_array_len do

    memo1.lines.add ( inttostr ( a[i] ) ) ;

End;

-------------------------------------------------------------

                     (* Insertion Sort *)

Procedure TForm1.Button4Click(SEnder: TObject);
Var
  b       : array [1..max_array_len] of byte ;

  Procedure shift_right ( pos , high : byte ) ;

    Var
      i : integer ;

    Begin
      for i := high downto pos do

        b[i+1] := b[i] ;

    End ;

  Function find_pos ( insert_value , high : byte ): byte ;

    Var
      i : integer ;

    Begin
      i := 1 ;

      while ( i <= high ) and ( b[i] > insert_value ) do

        inc ( i ) ;

      result := i ;

    End ;

Var
  i   : integer ;

  pos : byte ;

Begin
  b[1] := a[1] ;

  for i := 2 to max_array_len do

    Begin
      pos := find_pos ( a[i] , i-1 ) ;

      shift_right ( pos , i-1 ) ;

      b[pos] := a[i] ;

    End ;

  // copy b to a

  for i := 1 to max_array_len do

    a[i] := b[i] ;

  // printing sorted list

  memo1.lines.add ( '-insertion sort result-' ) ;

  for i :=  1 to max_array_len do

    memo1.lines.add ( inttostr ( a[i] ) ) ;

End;

--------------------------------------------------------------

                          (* Bubble Sort *)

Procedure TForm1.Button5Click(SEnder: TObject);

  Procedure swap ( Var a,b : byte ; Var s1,s2 : str_30 ) ;

    Var
      temp  : byte ;

      temps : str_30 ;

    Begin
      temp := a ;

      a := b ;

      b := temp ;

      temps := s1 ;

      s1 := s2 ;

      s2 := temps ;

    End ;

Var
  i,j : integer ;

Begin
  for j := 1 to max_array_len-1 do

    for i := 1 to max_array_len-j do

      if a[i] < a[i+1] then

        swap ( a[i] , a[i+1] , names[i] , names[i+1]) ;

  // printing sorted list

  memo1.lines.add ( '-bubble sort result-' ) ;

  for i :=  1 to max_array_len do

    memo1.lines.add ( names[i]+' :'+inttostr ( a[i] ) ) ;

End;
13-4- جستجوي خطي و باينري

براي جستجوكردن در يك ليست مانند يك آرايه، روشهاي مختلفي وجود دارد. 
جستجوي خطي و باينري، دو نمونه از اين روشها هستند. جستجوي خطي در آرايه​هاي نامرتب، و جستجوي باينري در آرايه​هاي مرتب صورت مي​گيرد.
جستجوي خطي - در اين روش جستجو، عنصر مورد جستجو، با تك تك عناصر آرايه مقايسه مي​شود. در صورت برابري، عمل مقايسه پايان مي​پذيرد. در غير اين صورت، اين روند براي عناصر ديگر آرايه انجام مي​شود. 

جستجوي باينري - اين روش بدين صورت است كه ابتدا عنصر مورد جستجو، با عنصر وسط آرايه مقايسه مي​شود. اگر برابر بود، جستجو با پايان مي​رسد، در غير اين صورت اگر عنصر مورد جستجو از عنصر وسط بزرگتر يا كوچكتر بود، آرايه به دو قسمت تقسيم مي​گردد. قسمت اول، قسمتي است كه عناصر موجود در آن بزرگتر از عنصر مورد جستجو هستند. قسمت دوم، قسمتي است كه عناصر موجود در آن كوچكتر از عنصر مورد جستجو هستند. اگر عنصر مورد جستجو از عنصر وسط بزرگتر بود، جستجو در قسمت بالايي آرايه، وگرنه جستجو در قسمت پائيني آرايه انجام مي​شود. اين روند تا يافتن عنصر مورد نظر يا بررسي كل عناصر آرايه ادامه پيدا مي​كند.
تمرين

برنامه مربوط به جستجوي خطي و باينري را بنويسيد.
13-5- كاراكترها و متغيرهاي كاراكتري

يك متغير كاراكتري، نوع متغيري است كه مي​تواند حروف (كاراكترها) را در خود نگهداري كند. حروف شامل حروف الفبا، ارقام و نشانه​ها است. يك كاراكتر يا از نوع كاراكتر ansi است كه حاوي كاراكتر بر طبق جدول ASCII است و يا يك كاراكتر wide است كه حاوي كاراكتر بر طبق جدول Unicode است.

كاراكتر ASCII- در جدول ASCII، هر كاراكتر داراي يك كد است. مثلا كاراكتر ‘A’ كد 64 و كاراكتر ‘B’ كد 65 و ... علت در نظر گرفتن كد مخصوص براي هر حرف اين است كه مي​دانيم تنها اعداد هستند كه در رايانه (البته به شكل صفر و يكشان) ذخيره مي​شوند. از اين رو براي ذخيره​سازي حروف، بايد آنها را به نوعي به يك عدد اختصاص داد تا بتوان آنها را توسط اعداد (كد) شان به رايانه شناساند. لذا هر حرف تبديل به يك عدد مي​شود. اين حروف به همراه كدشان در جدولي با نام جدول ASCII در ضميمه اين كتاب آمده​اند. با مراجعه به اين جدول مي​توان كد هر كاراكتر را پيدا نمود. جدول ASCII حاوي كاراكترها با كدهاي 0 الي 255 است كه كد 0 مربوط به null، كدهاي 1 الي 32 مربوط به كاراكترهاي كنترلي، كدهاي 33 الي 127 براي كاراكترهاي حروف لاتين و علائم و كدهاي 128 الي 255 متعلق به كاراكترهاي توسعه (مثلا فارسي) است.

يك كاراكتر ASCII در يك بايت ذخيره مي​شود. مي​دانيم كه با يك بايت قادر به ساخت 256 عدد مختلف هستيم. كد 0 تا 127 (128 كد اول) در ASCII مربوط به كاراكترهاي استاندارد و كدهاي 128 تا 255 (128 كد بعدي) كاراكترهاي قابل تعريف براي هر زبان است. مثلا براي هر حرف در زبان فارسي، يك كد بين 128 تا 255 در نظر گرفته شده است. اما مشكل اينجاست كه 128 كد دوم در ASCII تنها در زبان تعريفي خود معنا پيدا مي​كند. مثلا در زبان فارسي، حروف فارسي كدهاي مخصوص به خود در محدوده 128 تا255 و در زبان آلماني، حروف آلماني كدهاي مخصوص به خود در محدوده 128 تا 255 را دارا هستند. از اين رو نمي​توان براي هر زبان كدهاي منحصر به فردي ايجاد نمود.

كاراكتر Unicode-  يوني​كد، با ظرفيتي براي بيش از يك ميليون كاراكتر، امكان كدگذاري كليه زبانهاي نوشتاري دنيا (از جمله فارسي) را فراهم مي​كند. يك كاراكتر يوني​كد در دو بايت ذخيره مي​شود. 128 كد اول در يوني​كد، همان 128 كد در ASCII هستند، بقيه كدها (از 128 تا 65535) تقريبا تمام زبانهاي دنيا را شامل مي​شود. با پيدايش يوني​كد ديگر مشكل گفته شده در ASCII وجود ندارد. Windows2000 و XP ، سيستم عامل​هاي طراحي شده بر مبناي يوني​كد هستند.

انواع متغيرهاي كاراكتري ASCII و Unicode موجود در زبان دلفي در جدول 1-13 آمده است:


	نوع متغير
	طول به بايت
	ansi/Unicode

	Char
	1
	ansi

	Ansichar
	1
	ansi

	Widechar
	2
	Unicode


يك كاراكتر ساده مي​تواند در يك متغير كاراكتري ذخيره شود. نحوه تعريف كاراكتر به شكل زير است:

                       Var
                         ch : char ;

                         ...

                         ch := ‘a’ ;

                         if ch = ‘a’ then

                         ...
13-6- رشته​ها و متغيرهاي رشته​اي
يك رشته، متغيري است كه مي​تواند يك عبارت (مجموعه​اي از كاراكترها) را در خود ذخيره كند. به عنوان مثال، از اين متغيرها مي​توان براي نگهداري اسامي استفاده نمود. انوع متغيرهاي رشته​اي در جدول 2-13 آمده است:

	نوع بكارگيري
	حافظه مورد نياز
	حداكثر طول
	نوع

	Backward compatibility
	2 تا 256 بايت
	255 كاراكتر
	shortstring

	8-bit (ANSI) characters
	4 تا 2 گيگابايت
	2431 كاراكتر
	ansistring

	ansichar
	
	3 مگابايت
	string

	Unicode characters
	4 تا 2گيگابايت
	2431 كاراكتر
	widestring


در اكثر برنامه​ها ما تنها از shortstring استفاده مي​كنيم. چرا كه طول آن براي اكثر رشته​هاي مورد نياز در برنامه مانند نام، نام خانوادگي و ... كافي است.


يك متغير رشته​اي به شكل زير تعريف مي​گردد:
Var
  S : string ;
براي ريختن يك عبارت در يك متغير رشته​اي كافي است به صورت زير عمل كنيم:
  s := ‘ali’ ;
به اين ترتيب، يك رشته با نام s تعريف شده است كه عبارت "ali" را در داخل آن قرار داده​ايم.
رشته​ها ماهيت آرايه​اي دارند و مي​توان با آنها مثل يك آرايه عمل كرد. بنابراين از روش مقداردهي در آرايه نيز مي​توان عبارتي را در يك متغير رشته​اي قرار داد. وليكن ابتدا بايد طول رشته را مشخص كرد. براي مشخص نمودن طول يك رشته كافي است تا اندازه مورد نظر را به همراه عملگر # به عضو ابتداي رشته نسبت دهيم. اين عملگر كاراكتر شماره پس از آن را برمي​گرداند. مثلا #3 يعني كاراكتر شماره 3. به مثال 13-3 توجه كنيد.
مثال 13-3

رشته​اي تعريف كرده و با روش مقداردهي آرايه​اي، عبارت ‘ali’  را در آن قرار دهيد.
Var
  S : shortstring ;

Begin
  S[0] := #3  ;

  S[1] := ‘a’ ;

  S[2] := ‘l’ ;

  S[3] := ‘i’ ;

End ;

13-6-1- الحاق رشته​ها
مي​توان دو رشته را با هم جمع نمود، منتهي اين جمع، جمع طولي است و به آن الحاق گويند. براي جمع نمودن دو رشته، از عملگر "+" استفاده مي​كنيم. به دستورات زير دقت نماييد:

S1 := ‘string1’ ;

S2 := ‘string2’ ;

S3 := S1 + S2 ;
در اين صورت مقدار S3 برابر خواهد با: string1string2  

نكته
مي​توان به جاي رشته، از يك آرايه​اي از كاراكترها استفاده نمود.
 مثلا: A : array[0..10] of char ;            
با اين تعريف، آرايه​اي 10 عضوي از كاراكتر با نام A ساخته شده است. اما اين آرايه را نمي​توان مانند يك رشته به صورت يكپارچه بكار برد. مثلا:

S : shortstring ;

A: array[1..10] of char ;

….

A[1] := ‘A’ ;

A[2] := ‘L’ ;

A[3] := ‘I’ ;

مي​توان نوشت:

S := ‘ALI’ ;
اما نمي​توان نوشت:
A := ‘ALI’ ;
همچنين مي​توان نوشت:
Edit1.text := S ;

اما نمي​توان نوشت:
Edit1.text := A ;
مثال13 -4

برنامه​اي بنويسيد كه با فشار دادن يك كليد، جدول ضربي را در يك memo نمايش دهد.

Var
  mat  : array [1..10,1..10] of byte ;

  i,j  : integer ;

  s    : shortstring ;

Begin
  (* compute matrix*)

  for i :=1 to 10 do

    for j := 1 to 10 do

      mat[i,j] := i*j ;

  (* print out matrix in memo *)

  for i := 1 to 10 do

    Begin (* print a row *)

      s := '' ;    (* s[0] := #0 *)

      for j := 1 to 10 do

        s := s + (inttostr ( mat[i,j] )) ;

      memo1.Lines.add ( s ) ;

    End ;

End;
با اجراي اين برنامه مشاهده مي​كنيد كه جدول به درستي كار مي​كند، اما به شكلي نامنظم. لازم است برنامه را به شكلي تغيير دهيم تا مابين اعداد به اندازه كافي فاصله ايجاد شود. بنابراين اين برنامه را با نوشتن تابعي كه طول يك رشته را به 4 كاراكتر برساند، بگونه​اي كامل مي​كنيم كه شكل جدول به صورت منظم درآيد. سپس اين تابع را قبل از اضافه نمودن هر عددي به جدول به كار مي​گيريم.

Procedure TForm1.BitBtn1Click(SEnder: TObject);

  Function set_len_str_to_4 ( s : shortstring ) : shortstring ;

     Begin
       while s[0] < #4 do

         s := ' ' + s ;

       result := s ;

     End ;

Var
  mat  : array [1..10,1..10] of byte ;

  i,j  : integer ;

  s    : shortstring ;

Begin
  (* compute matrix*)

  for i :=1 to 10 do

    for j := 1 to 10 do

      mat[i,j] := i*j ;

  (* print out matrix in memo *)

  for i := 1 to 10 do

    Begin (* print a row *)

      s := '' ;    (* s[0] := #0 *)

      for j := 1 to 10 do

        s := s + set_len_str_to_4 (inttostr ( mat[i,j] )) ;

      memo1.Lines.add ( s ) ;

    End ;

End;
تمرين

تابع set_len_str_to_4 را بگونه​اي بنويسيد كه هر رشته را با افزايش فضاي خالي يا كاهش (يا حذف از سمت راست) به طول مشخصي برساند.
مثال 13-5

يك برنامه بنويسيد كه تمام كاراكترهاي كوچك يك رشته را به كاراكترهاي بزرگ تبديل كند.

براي اين كار از تابع upcase استفاده مي​كنيم.
13-6-2- تابع UpCase و تابع UpperCase
تابع UpCase، يك كاراكتر كوچك را به يك كاراكتر بزرگ تبديل مي​كند، و تابع UpperCase،  رشته​اي را خوانده و تمام كاراكترهاي كوچك آن را به كاراكترهاي بزرگ تبديل مي​كند.

Function upcase_str ( s : shortstring ) : shortstring ;

  Var
    i : integer ;

  Begin
    for i := 1 to length ( s ) do

      s[i] := upcase ( s[i] ) ;

    result := s ;

  End ;

مثال 13-6

يك تابع بنويسيد كه كليه فاصله​هاي يك رشته را حذف كند.
Function remove_str_space ( s : shortstring ) : shortstring ;

Var
  i       : integer ;

  new_len : byte ;

Begin
  new_len := 0 ;

  for i := 1 to length ( s ) do

    if s[i] <> ' ' then

      Begin
        inc ( new_len ) ; (* new_len := new_len + 1 *)

        result[new_len] := s[i] ;

      End ;

  result[0] := chr ( new_len ) ;

End;

مثال 13-7

يك تابع بنويسيد كه تمام فاصله​هاي متوالي بيش از دو كاراكتر را به يك فاصله يك كاراكتري تبديل كند.
Function less_str_space ( s : shortstring ) : shortstring ;

Var
  i         : integer ;

  new_len   : byte ;

  prev_spc  : boolean ;

Begin
  new_len := 0 ;

  prev_spc := False ;

  for i := 1 to length ( s ) do

    if s[i] <> ' ' then

      Begin
        inc ( new_len ) ; (* new_len := new_len + 1 *)

        result[new_len] := s[i] ;

        prev_spc := False ;

      End
    else (* s[i] = ' ' *)

      if not prev_spc then

        Begin (* write first space *)

          inc ( new_len ) ; (* new_len := new_len + 1 *)

          result[new_len] := s[i] ;

          prev_spc := true ;

        End ;

  result[0] := chr ( new_len ) ;

End;
مثال 13-8

تابعي بنويسيد كه دو رشته را با حذف فواصل و كوچك و بزرگ كردن حروف لاتين با هم مقايسه كند.

Function compare_strings ( s1,s2 : shortstring ) : boolean ;

  Begin
    s1 := remove_str_space ( uppercase ( s1 ) ) ;

    s2 := remove_str_space ( uppercase ( s2 ) ) ;

    if s1 = s2 then

      result := true       (*Equal*)

    else

      result := False ;   (*Not Equal*)

  End ;

نكته

فرض كنيد رشته​اي از نوع shortstring تعريف نموده​ايم. همان​طوركه مي​دانيم اين رشته مي​تواند حاوي حداكثر 255 كاراكتر باشد. حال فرض كنيد كه در برنامه مقدار ‘ali’ را به متغير رشته​اي داده​ايم. حال اگر رشته را چاپ نماييم، تنها عبارت ‘ali’ چاپ خواهد شد و نه چيز ديگري، در صورتي كه طول رشته تعريفي ما 255 كاراكتر است. ولي اين سؤال مطرح است كه چگونه اين متغير دانسته كه ما تنها از 3 كاراكتر اين رشته استفاده نموده​ايم، و تنها 3 كاراكتر بكارگرفته شده را چاپ نموده است.

در جواب به اين سؤال بايد متذكر شويم كه در رشته تعريفي، بايت صفرم حاوي طول پر شده رشته مربوطه است. اين بايت حاوي كاراكتري است كه كد آن بيانگر طول رشته است. مثلا در يك متغير از نوع shortstring كه مقدار ‘ali’  را به آن داده​ايم، در خانه صفرم اين متغير، كاراكتري قرار دارد كه كد آن 3 است. توسط تابع length مي​توان مقدار طول رشته كه همان بايت صفرم رشته است را مشاهده نمود. به مثال زير توجه كنيد:

Var
  S : shortstring ; 

  I : byte ;

…

  S := ‘ali’ ;

  I := length(s) ;

…

در اين مثال مقدار متغير I، 3 خواهد شد. 
13-6-3- اپراتورهاي # و chr
اين دو اپراتور مقدار عدد را به كاراكتر تبديل مي​كنند. مثلا در رشته​اي كه مقدار ‘ali’ را در آن قرار داده​ايم، عبارت زير يك عبارت صحيح است:

If s[0] = #3 then

 …
چرا كه مقدار خانه صفر، حاوي كاراكتر است كه كدش مقدار 3 است. يكي از كاربردهاي عملگر # خالي نمودن رشته است، بدين صورت كه خانه صفرم آرايه را برابر صفر قرار دهيم. بدين ترتيب طول رشته صفر در نظر گرفته مي​شود. مثلا براي رشته​اي با نام S، دستور زير آن را تهي مي​سازد. 

s[0] := #0
البته براي تهي نمودن يك رشته راه ديگري نيز وجود دارد، به صورت زير: 

s := '' ;      
#64 = char(64) = chr(64) = 'A' ;
نكته1: 

فرق بين تابع chr و # چيست؟ در تابع chr، مي​توان يك متغير را قرار داد ولي در # نمي​توان! مثلا مي​توان نوشت:

i := 5 

… chr(i) …

اما نمي​توان نوشت:

…#i …
نكته 2:
به نظر شما فرق بين char(64) با chr(64) چيست! پاسخ اين سؤال را كمي جلوتر در مفهوم TypeCasting خواهيم دارد.

13-6-4- PChar
نوع ديگري از رشته​ها و عبارات وجود دارد كه به جاي آنكه طول رشته در ابتداي آن ذخيره شود، در انتهاي رشته كاراكتر صفر( null ) قرار مي​گيرد. به اين نوع از رشته​ها null terminated strings  گفته مي​شود. char اين نوع از رشته​ها است. در اين كتاب با pchar عمليات رشته​ها را انجام نمي​دهيم. توابع عمليات pchar  با توابع string معمولي متفاوت است. 

13-7- Type Casting
در صورتي كه بخواهيم يك متغير از يك نوع را به يك متغير از نوع ديگر منتقل كنيم، مي​توانيم از تبديل قالب نوع، يا type cast استفاده كنيم. براي اين كار از عنوان خود نوع مورد نظر استفاده مي​كنيم. مثلا integer('a') يعني اينكه كاراكتر a به شكل integer درآورده شود. البته هرجايي نمي​توان از typecast استفاده نمود. بنابراين فرق chr(64) با char(64) در اين است كه chr تابعي است كه ورودي آن از نوع بايت و خروجي آن كاراكتر است. ولي char(64) يعني اينكه 64 تبديل به كاراكتر شود. مثال زير نيز چند نوع از اين تبديل را نشان مي​دهد:

i := integer ;

b : Boolean ;

c : char ;

i := integer(b) ;

b := Boolean(c) ;

c := char(48) ;
i := byte(‘w’) ;
13-8- 
نوشتن توضيحات در برنامه

برخي اوقات لازم است تا توضيحاتي در متن برنامه نوشته شود. توجه داشته باشيد كه متن برنامه بايد قابل خواندن و درك توسط كامپايلر باشد. ولي توضيحاتي كه ما مي​نويسيم، براي كامپايلر نيست. بلكه براي خودمان و يا براي برنامه​نويس ديگري كه برنامه ما را مي​خواند است. تعجب نكنيد، ما لازم است براي خودمان هم توضيحاتي را بنويسيم. شما چرا مطالب درسها را يادداشت مي​كنيد؟ چون ممكن است آنها را فراموش كنيد. نكاتي كه در برنامه شما وجود دارد هم ممكن است فراموش شود. وقتي با تنها 10 خط برنامه سروكار داشته باشيد، مسئله زياد مهم نيست، اما وقتي صحبت از چند ده هزار خط و يا چند صد هزار خط برنامه مي​شود، ديگر اينكه فلان جا چكار كرديد و چرا اين دستور را اينجا گذاشتيد به سادگي به ياد شما نخواهد آمد. براي همين بايد بتوانيم توضيحاتي را در داخل متن برنامه بگنجانيم.


علاوه بر آنكه براي خوانايي برنامه بايد از اسامي با مسمي و طريقه صحيح نوشتن متن برنامه استفاده كنيم، يك برنامه بايد داراي توضيحات كافي باشد. توضيحات به درك عناصر و ساختمان برنامه كمك مي​كند. بخصوص وقتي كه مدتي از نوشتن برنامه گذشته باشد؛ ولي توضيحات حتي در زمان نوشتن برنامه هم كمك كننده است.

روشهاي مختلف نوشتن توضيحات در برنامه به سه طريق زير است:

· (* it is a simple comment*)

· { it is a comment also - it used for temporary blocking}

· // it is a one line only comment

// it used for quick and abstract comments      

خصوصيات توضيحات:

· موجز باشد. زياده​گويي نكند.
· صريح و شفاف باشد.
· بيش از حد نباشد (تعداد توضيحات) (هر خط توضيح احتياج ندارد -  فقط در مكانهاي مهم و كليدي)
به عنوان نمونه يكي از قسمتهاي برنامه كه معمولا نياز به نوشتن توضيحات دارد، رويه​ها است. در ابتداي رويه بهتر است توضيح دهيم كه اين رويه تعريف شده چه عملي را انجام مي​دهد.
13-9- رنگها در محيط ويرايشي

ويراستار محيط برنامه اين امكان را مي​دهد كه رنگ عناصر مختلفي مانند شناسه​ها، ارقام، توضيحات و .. را تعيين كنيد. اين تعيين رنگ نه براي زيبايي است، بلكه به وضوح و شفافيت متن و سرعت تشخيص عناصر كمك مي​كند. تعيين رنگهاي ويرايش در بخش زير انجام مي​شود:
Tools/editor option/color
شكل 13-8 اين محيط را نشان مي​دهد. 

براي تفكيك​پذيري بهتر و نيز توجه بصري به اهميت و جايگاه عناصر، استفاده از رنگهاي زير پيشنهاد مي​شود:

· توضيحات – سبز سير (ايتاليك)
· كلمات ذخيره شده – بنفش سير (پررنگ)
· شناسه​ها – سياه 
· علائم – قرمز روشن
· رشته​ها – فيروزه​اي سير
· اعداد – آبي روشن
13-10- چگونگي نوشتن متن برنامه

يك برنامه رايانه​اي، جزء پيچيده​ترين ساخته​هاي بشر محسوب مي​شود. طول برخي از برنامه​هاي رايانه​اي، به بيش از يك ميليون خط مي​رسد. و مي​دانيم كه در صورتي كه اشتباهي كوچك در اين يك ميليون خط بروز كند، ممكن است كل برنامه دچار آسيب جدي شود.

پس در نوشتن برنامه، متن آن بايد به گونه​اي نوشته شود كه سازماندهي مناسبي داشته باشد و با نظم ايجاد شود. اين نظم به نوشتن و يافتن خطاها و اصلاحات كمك بزرگي مي​كند.
ساختمان برنامه بايد داراي نظم باشد. استفاده از نامهاي بامعنا و به جا، استفاده از توضيحات و ... لازم است، ولي كافي نيست. ساختمان خود برنامه بايد داراي نظم مناسب باشد.

شكل 13-8- محيط Editor Properties براي تنظيم خصوصيات مختلف متن برنامه، مانند رنگ 





شكل 13-7- فرم برنامه مثال 13-2

شكل 13-5- مثالي از يك مرتب�سازي درجي





شكل 13-4- مثالي از يك مرتب�سازي انتخابي





شكل 13-3- پنجره Project Manager براي مديريت فرمها و واحدهاي موجود در برنامه





شكل 13-2- فرم دوم برنامه مثال 13-2





شكل 13-1- فرم اول برنامه مثال 13-2

